pragma solidity ^0.4.21;

/**
 * @title SafeMath
 * @dev Math operations with safety checks that throw on error
 */
library SafeMath {

  /**
  * @dev Multiplies two numbers, throws on overflow.
  */
  function mul(uint256 a, uint256 b) internal pure returns (uint256 c) {
    if (a == 0) {
      return 0;
    }
    c = a * b;
    assert(c / a == b);
    return c;
  }

  /**
  * @dev Integer division of two numbers, truncating the quotient.
  */
  function div(uint256 a, uint256 b) internal pure returns (uint256) {
    // assert(b > 0); // Solidity automatically throws when dividing by 0
    // uint256 c = a / b;
    // assert(a == b * c + a % b); // There is no case in which this doesn't hold
    return a / b;
  }

  /**
  * @dev Subtracts two numbers, throws on overflow (i.e. if subtrahend is greater than minuend).
  */
  function sub(uint256 a, uint256 b) internal pure returns (uint256) {
    assert(b <= a);
    return a - b;
  }

  /**
  * @dev Adds two numbers, throws on overflow.
  */
  function add(uint256 a, uint256 b) internal pure returns (uint256 c) {
    c = a + b;
    assert(c >= a);
    return c;
  }
}

/**
 * @title ERC20Basic
 * @dev Simpler version of ERC20 interface
 * @dev see https://github.com/ethereum/EIPs/issues/179
 */
contract ERC20Basic {
  function totalSupply() public view returns (uint256);
  function balanceOf(address who) public view returns (uint256);
  function transfer(address to, uint256 value) public returns (bool);
  event Transfer(address indexed from, address indexed to, uint256 value);
}

/**
 * @title ERC20 interface
 * @dev see https://github.com/ethereum/EIPs/issues/20
 */
contract ERC20 is ERC20Basic {
  function allowance(address owner, address spender) public view returns (uint256);
  function transferFrom(address from, address to, uint256 value) public returns (bool);
  function approve(address spender, uint256 value) public returns (bool);
  event Approval(address indexed owner, address indexed spender, uint256 value);
}

/**
 * @title Basic token
 * @dev Basic version of StandardToken, with no allowances.
 */
contract BasicToken is ERC20Basic {
  using SafeMath for uint256;

  mapping(address => uint256) balances;

}

/**
 * @title Standard ERC20 token
 *
 * @dev Implementation of the basic standard token.
 * @dev https://github.com/ethereum/EIPs/issues/20
 * @dev Based on code by FirstBlood: https://github.com/Firstbloodio/token/blob/master/smart_contract/FirstBloodToken.sol
 */
contract StandardToken is ERC20, BasicToken {

  mapping (address => mapping (address => uint256)) internal allowed;


  /**
   * @dev Approve the passed address to spend the specified amount of tokens on behalf of msg.sender.
   *
   * Beware that changing an allowance with this method brings the risk that someone may use both the old
   * and the new allowance by unfortunate transaction ordering. One possible solution to mitigate this
   * race condition is to first reduce the spender's allowance to 0 and set the desired value afterwards:
   * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
   * @param _spender The address which will spend the funds.
   * @param _value The amount of tokens to be spent.
   */
  function approve(address _spender, uint256 _value) public returns (bool) {
    allowed[msg.sender][_spender] = _value;
    emit Approval(msg.sender, _spender, _value);
    return true;
  }

  /**
   * @dev Increase the amount of tokens that an owner allowed to a spender.
   *
   * approve should be called when allowed[_spender] == 0. To increment
   * allowed value is better to use this function to avoid 2 calls (and wait until
   * the first transaction is mined)
   * From MonolithDAO Token.sol
   * @param _spender The address which will spend the funds.
   * @param _addedValue The amount of tokens to increase the allowance by.
   */
  function increaseApproval(address _spender, uint _addedValue) public returns (bool) {
    allowed[msg.sender][_spender] = allowed[msg.sender][_spender].add(_addedValue);
    emit Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

  /**
   * @dev Decrease the amount of tokens that an owner allowed to a spender.
   *
   * approve should be called when allowed[_spender] == 0. To decrement
   * allowed value is better to use this function to avoid 2 calls (and wait until
   * the first transaction is mined)
   * From MonolithDAO Token.sol
   * @param _spender The address which will spend the funds.
   * @param _subtractedValue The amount of tokens to decrease the allowance by.
   */
  function decreaseApproval(address _spender, uint _subtractedValue) public returns (bool) {
    uint oldValue = allowed[msg.sender][_spender];
    if (_subtractedValue > oldValue) {
      allowed[msg.sender][_spender] = 0;
    } else {
      allowed[msg.sender][_spender] = oldValue.sub(_subtractedValue);
    }
    emit Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

  /**
  * @dev Function to check the amount of tokens that an owner allowed to a spender.
  * @param _owner address The address which owns the funds.
  * @param _spender address The address which will spend the funds.
  * @return A uint256 specifying the amount of tokens still available for the spender.
  */
  function allowance(address _owner, address _spender) public view returns (uint256) {
    return allowed[_owner][_spender];
  }

}

/**
 * @title Partial Basic
 * @dev Partial Basic is an experimental cryptocurrency that guarantees
 * unconditional, indeterminate rewards to network participants.
 */
contract PartialBasic is StandardToken {
  using SafeMath for uint256;

  string public constant name = "Partial Basic"; // solium-disable-line uppercase
  string public constant symbol = "PB"; // solium-disable-line uppercase
  uint8 public constant decimals = 18; // solium-disable-line uppercase

  uint256 public constant BASE_REWARD = 20000 ether;
  uint256 private constant PRECISION = 10**18;

  uint256 public totalNodes;
  uint256 public rewardStartTime;
  uint256 public rewardCheckpoint;
  uint256 private rewardTimestamp;

  mapping(address => uint256) public nodes;
  mapping(address => uint256) private claimed;

  event Mint(address indexed to, uint256 amount);
  event AddNode(address indexed owner);

  /**
  * @dev add a node for a specified address.
  * @param _owner The address to add a node for.
  */
  function addNode(address _owner) external {
    uint256 checkpointCandidate;

    if (rewardStartTime == 0) {
      // initialise rewards
      rewardStartTime = block.timestamp;
    } else {
      // reward per node must increase to be a valid checkpoint
      // or a valid reward for this block must have already been checkpointed
      checkpointCandidate = rewardPerNode();
      require(checkpointCandidate > rewardCheckpoint || block.timestamp == rewardTimestamp);
    }

    // claim outstanding rewards
    sync(_owner);

    if (rewardCheckpoint != checkpointCandidate) {
      // checkpoint the total reward per node
      rewardCheckpoint = checkpointCandidate;
    }

    if (rewardTimestamp != block.timestamp) {
      // reset the timestamp for the reward
      rewardTimestamp = block.timestamp;
    }

    // add node for address
    nodes[_owner] = nodes[_owner].add(1);

    // prevent new nodes from claiming old rewards
    claimed[_owner] = rewardCheckpoint.mul(nodes[_owner]);

    // update the total nodes in the network
    totalNodes = totalNodes.add(1);

    // fire event
    emit AddNode(_owner);
  }

  /**
  * @dev Gets the total reward for a node.
  * @return A uint256 representing the total reward of a node.
  */
  function rewardPerNode() public view returns (uint256) {
    // no reward if there are no active nodes
    if (totalNodes == 0) {
      return;
    }

    // days since last checkpoint
    uint256 totalDays = block.timestamp.sub(rewardTimestamp).mul(PRECISION).div(1 days);

    // reward * days / nodes
    uint256 newReward = BASE_REWARD.mul(totalDays).div(PRECISION).div(totalNodes);

    // checkpoint + newReward
    return rewardCheckpoint.add(newReward);
  }

  /**
  * @dev Gets the outstanding reward of a specified address.
  * @param _owner The address to query the reward of.
  * @return A uint256 representing the outstanding reward of the passed address.
  */
  function calculateReward(address _owner) public view returns (uint256) {
    // address must be running a node
    if (isMining(_owner)) {
      // return outstanding reward
      uint256 reward = rewardPerNode().mul(nodes[_owner]);
      return reward.sub(claimed[_owner]);
    }
  }

  /**
  * @dev sync an outstanding reward for a specified address.
  * @param _owner The address to sync rewards for.
  */
  function sync(address _owner) public {
    uint256 reward = calculateReward(_owner);
    if (reward > 0) {
      claimed[_owner] = claimed[_owner].add(reward);
      balances[_owner] = balances[_owner].add(reward);
      emit Mint(_owner, reward);
      emit Transfer(address(0), _owner, reward);
    }
  }

  /**
  * @dev transfer token for a specified address.
  * @param _to The address to transfer to.
  * @param _value The amount to be transferred.
  */
  function transfer(address _to, uint256 _value) public returns (bool) {
    sync(msg.sender);
    require(_to != address(0));
    require(_value <= balances[msg.sender]);

    balances[msg.sender] = balances[msg.sender].sub(_value);
    balances[_to] = balances[_to].add(_value);
    emit Transfer(msg.sender, _to, _value);
    return true;
  }

  /**
  * @dev Transfer tokens from one address to another.
  * @param _from address The address which you want to send tokens from.
  * @param _to address The address which you want to transfer to.
  * @param _value uint256 the amount of tokens to be transferred.
  */
  function transferFrom(address _from, address _to, uint256 _value) public returns (bool) {
    sync(_from);
    require(_to != address(0));
    require(_value <= balances[_from]);
    require(_value <= allowed[_from][msg.sender]);

    balances[_from] = balances[_from].sub(_value);
    balances[_to] = balances[_to].add(_value);
    allowed[_from][msg.sender] = allowed[_from][msg.sender].sub(_value);
    emit Transfer(_from, _to, _value);
    return true;
  }

  /**
  * @dev Gets the balance of the specified address.
  * @param _owner The address to query the the balance of.
  * @return A uint256 representing the amount owned by the passed address.
  */
  function balanceOf(address _owner) public view returns (uint256) {
    return balances[_owner].add(calculateReward(_owner));
  }

  /**
  * @dev returns the approximate total number of tokens in existence
  * @return A uint256 representing the approximate total number of tokens in existence.
  */
  function totalSupply() public view returns (uint256) {
    if (rewardStartTime == 0) {
      return;
    }

    // days since start of rewards
    uint256 totalDays = block.timestamp.sub(rewardStartTime).mul(PRECISION).div(1 days);

    // reward * days
    return BASE_REWARD.mul(totalDays).div(PRECISION);
  }

  /**
  * @dev returns the mining status of the passed address.
  * @return A uint256 representing the mining status of the passed address.
  */
  function isMining(address _owner) public view returns (bool) {
    return nodes[_owner] != 0;
  }

  /**
  * @dev A batch query to get all node information for a specified address.
  * @param _owner The address to query the details of.
  * @return A bool representing the mining status of the passed address.
  * @return A uint256 representing the number of nodes owned by the passed address.
  * @return A uint256 representing the amount owned by the passed address.
  * @return A uint256 representing the outstanding reward of the passed address.
  * @return A uint256 representing the total reward per node.
  * @return A uint256 representing the total nodes in the network.
  * @return A uint256 representing the total number of tokens in existence.
  */
  function getInfo(address _owner) public view returns (bool, uint256, uint256, uint256, uint256, uint256, uint256) {
    return (isMining(_owner), nodes[_owner], balanceOf(_owner), calculateReward(_owner), rewardPerNode(), totalNodes, totalSupply());
  }

}